Пушпышев А.
Краткое руководство для школьников

по правлению операционной системой без использования графической оболочки с помощью пакетных файлов
Оглавление
Введение










3
История










3
Командный процессор







3
Командная строка и команды






4
Текущий каталог. Абсолютные и относительные пути


5
Командные файлы








5
Комментирование командного файла и его выдачи. 
Команды echo и rem








6
Передача командному файлу параметров




8
Переменные









9
Условные операторы








11
Массовая обработка файлов






13
Передача управления другому командному файлу


14
Вопросы и упражнения для самопроверки




16
Введение

В операционной системе MS-DOS и во всех операционных системах семейства Microsoft Windows простейшим средством автоматизации обработки файлов (и каталогов) служат так называемые командные файлы. Командные файлы — не единственное средство автоматизации работы пользователя даже из числа встроенных.  В операционных системах семейства Microsoft Windows, начиная с версии 98, имеется еще Windows Script(ing) Host. Для этих же целей можно использовать какой-нибудь интерпретируемый язык Perl, Python или Ruby. Однако перечисленные средства, хотя и мощны, требуют умения программировать, т.е. составлять и, что немаловажно, отлаживать программы, хотя бы на неплохом любительском уровне. Использование же командных файлов доступно всякому без особенной подготовки.

История

Программирование пакетных файлов появилось в MS-DOS и Windows с самого зарождения этих операционных систем. Коммандный интерпретатор в MS-DOS имеет название COMMAND.COM. Эволюция этой ветви пакетного программирования прошла через различные версии MS-DOS к Windows 95, Windows 98 и, наконец, к Windows Me. Наиболее известным пакетным файлом в этих системах является AUTOEXEC.BAT, специальный пакетный файл, который исполняется во время загрузки операционной системы. Новые версии Windows — Windows 2000, XP и Vista основаны не на MS-DOS, а на Windows NT. NT-подобные системы включают интерпретатор cmd.exe, который частично совместим с COMMAND.COM. Некоторые старые возможности MS-DOS недоступны, однако вместо них появились дополнительные возможности и команды. COMMAND.COM до сих пор включается в NT-подобные системы для обеспечения лучшей обратной совместимости. Существуют различные другие командные интерпретаторы, разработанные не компанией Microsoft и предоставляющие расширенный синтаксис команд для пакетного программирования. Примером может служить 4DOS.

Командный процессор

В состав многих операционных систем, в том числе, разработанных корпорацией Microsoft, входит командный процессор. Так называется программа, которая инициирует выполнение всевозможных действий в ответ на команды, вводимые пользователем с клавиатуры. В основном эти действия заключаются в запуске нужных программ с теми или иными параметрами. Но не только; дальше мы увидим, что некоторые команды выполняются непосредственно командным процессором. В основном это те команды, которые служат для управления контекстом и последовательностью выполнения команд. Однако мы не будем особенно глубоко задумываться о природе команд, во всяком случае, без необходимости. Важнее другое: любая программа, которую технически возможно запустить из командной строки, рассматривается командным процессором как команда. Он не делает различий между «родными» командами, изначально встроенными в операционную систему, и программами, которые были установлены поверх нее.
Командная строка и команды

Окно командного процессора в первозданном виде выглядит мрачно, и работать с ним большинству людей не очень удобно. Намного проще пользоваться файловыми менеджерами в стиле Norton Commander. В них предусмотрены как средства для быстрой навигации по файловой системе, так и срока для ввода команд.

Для того чтобы ввести команду:

· Наберите текст команды в командной строке.

· Нажмите на клавишу Enter.

Командный процессор и команды операционной системы описаны в эксплуатационной документации. Эта документация частично содержится внутри самой операционной системы. Для доступа к ней предназначена команда help. По этой команде на экран выводится список доступных команд. Для того чтобы получить описание конкретной команды, в качестве параметра команда help следует указать ее имя. На пример вот эта командная строка  выводит на экран описание команды for :  help for. Но результат  работы (т.н. выдача) команды help  не умещается на один экран. Та же проблема возникает с текстом описания команды for. Однако выдачу можно перенаправить в файл. Вот эта   командная строка формирует файл commands.txt, содержащий список всех команд MS-DOS:  help > commands.txt. Для того чтобы сформировать файл с описанием команды for, надо дать такую команду (имя выходного файла можете сделать любым): help for> for.txt. Всего в современных операционных системах корпорации Microsoft чуть меньше 80 команд, и чтобы рассказать про каждую требуется много времени. 

Несколько команд, полезных для автоматизации обработки файлов и  как ими пользоваться. Эти команды будут в дальнейшем использоваться в примерах. Детали можно уточнить по команде help:

· copy — копирование одного или нескольких файлов;

· del — удаление одного или нескольких файлов;

· move — перемещение одного или нескольких файлов или каталогов;

· rename (сокращенно ren) — переименование одного или нескольких файлов или каталогов;

· xcopy — копирование дерева подкаталогов;

· mkdir (сокращенно md) — создание каталога;

· rmdir (сокращенно rd) — удаление каталога.

Одно из общих правил синтаксиса команд MS-DOS состоит в том, что при указании параметров сначала указывается источник, а потом результат. Например, если мы хотим переместить файл file1.txt из каталога box в каталог table, мы должны ввести команду: move box\file1.txt table. Сначала что переместить, потом куда переместить.  Если нужно переименовать файл lena.txt в файл natasha.txt, то команда должна быть записана вот так: ren lena.txt natasha.txt. Сначала что переименовать, потом во что переименовать.
Текущий каталог. Абсолютные и относительные пути
При работе с файловыми командами исключительную важность приобретает понятие текущего каталога. Дело в том, что при указании файла в качестве параметра команды мы всегда используем один из двух возможных способов указания на них: либо абсолютный (полный) путь, либо относительный путь. В полном пути мы указываем все, начиная с диска (или сетевого имени компьютера), например d:\misha\box\car.txt. Какой бы каталог ни оказался текущим в момент ввода команды, полный путь будет соответствовать одному и тому же файлу. Для относительного пути текущий каталог служит отправной точкой. Простейший случай относительного пути — имя файла. В контексте выполнения команды оно означает файл с таким именем, расположенный в текущем каталоге. Для записи относительного пути к текущему каталогу существует условная запись . (точка). Для записи относительного пути к каталогу, в котором содержится текущий каталог, существует условная запись .. (две точки). Команда, показанная далее, копирует все файлы из текущего каталога в каталог neighbour, расположенный рядом с ним: copy *.* .\neighbour.
Командные файлы

При обработке большого количества файлов или при систематическом выполнении одних и тех же команд это становится обременительно. Поэтому в командном процессоре предусмотрена возможность выполнения командных файлов. Командный файл — это текстовый файл, в котором набраны команды (или хотя бы одна команда). Пример командного файла показан далее: 

help copy > copy.help  // Копирование данных о функции copy в текстовый файл copy.help
help move > move.help // Копирование данных о функции move в текстовый файл move.help
md msdos-help //Создание директории msdos-help
move *.help msdos-help //Перемещение всех файлов с типом .help в директорию msdos-help
Командным файлам принято давать расширение bat. По нему файлы этого типа опознаются командным процессором. Этот файл можно назвать, например, make-help.bat.

Для того чтобы запустить командный файл: 

· Введите его имя в качестве команды. После этого командный файл будет выполнен.

В командном файле каждая команда занимает одну строку. Точнее, существует способ расположить одну команду на нескольких подряд идущих строках, для этого непосредственно перед каждым переводом строки следует поставить символ «крышка» ^. (Необходимо, чтобы каждая «крышка» была последним символом в своей строке; после нее не должно быть пробелов и табуляций). Пример такой команды: 


if exist disser.txt ^

copy disser.txt ^

d:\science\papers\drafts\sources
Для простоты, чтобы каждый раз не делать оговорок, считают, что в некотором смысле это одна длинная «логическая» строка. При выполнении командного файла командный процессор просматривает его сверху вниз от первой строки к последней и выполняет команды в том порядке, в котором их обнаруживает. Выполняет он в целом их так, как если бы каждая строка была введена вручную. В целом, потому что некоторые команды при вводе вручную и при выполнении из командного файла ведут себя немного по-разному. При необходимости последовательность выполнения команд можно изменить с помощью управляющих команд. 

Комментирование командного файла и его выдачи. 
Команды echo и rem

Командный файл, по существу, представляет собой программу, написанную на языке командного процессора операционной системы. Текст программы полагается снабжать комментариями, чтобы, вернувшись к нему некоторое время спустя, не вспоминать мучительно, для чего эта программа нужна, и как она устроена.

В системе команд MS-DOS для оформления комментариев предусмотрена команда rem. Это фиктивная команда, которая не предполагает выполнения каких бы то ни было действий, но позволяет написать в строке после своего имени произвольный текст. Причем командный процессор не воспринимает его как синтаксическую ошибку. Пример оформления командного файла комментариями:

rem *****************************************************

rem Формирование файлов справки по командам copy и move

rem *****************************************************

rem Формируем файлы справки

help copy > copy.help

help move > move.help

rem Создаем каталог для хранения файлов справки

md msdos-help

rem Перемещаем файлы справки в подготовленный каталог

move *.help msdos-help

При выполнении приведенного выше командного файла все команды будут выводиться на экран по мере их выполнения, что не всегда удобно. Выдачу команд можно отключить с помощью команды @echo off. Символ «собака» перед командой echo означает, то и сама эта команда должна выполняться в «молчаливом» режиме. С таким же успехом мы могли бы не пользоваться командной echo off, а поместить «собаку» перед каждой командой. Во многих случаях требуется, чтобы командный файл выводил на экран (или в файл) те или иные сообщения. В одних случаях это могут быть сообщения об ошибках, в других информационные сообщения, объясняющие пользователю командного файла, что происходит в данный момент, когда-то мы формируем командным файлом какой-нибудь другой полезный файл. Для вывода сообщений применяется та же самая команда echo. В качестве параметра ей передают текст выводимого сообщения. Вот усовершенствованный текст программы: 

@echo off

rem *****************************************************

rem Формирование файлов справки по командам copy и move

rem *****************************************************

@echo Формируем файлы справки.Подождите...

rem Формируем файлы справки

help copy > copy.help
help move > move.help
rem Создаем каталог для хранения файлов справки

md msdos-help

rem Перемещаем файлы справки в подготовленный каталог

move *.help msdos-help

echo Готово!

Передача командному файлу параметров

Пусть нужно создать командный файл, который сначала формирует справку с описанием заданной пользователем команды, а потом загружает его для просмотра в блокнот. Фокус в том, чтобы при очередном запуске командного файла каким-то образом сообщить ему, какая именно команда нас интересует на этот раз.

Для решения этой задачи предусмотрен механизм обработки параметров. Работает он довольно просто. Если при запуске командного файла пользователь указал несколько параметров, то в тексте командного файла первый из них мы обозначаем записью %1, второй записью %2, третий записью %3 и т.д. Этими обозначениями используются в тексте командного файла примерно так же, как в естественной речи местоимениями. Текст командного файла, решающего поставленную задачу, приведен далее:

@echo off

rem Формируем файл с описанием команды, 

rem имя которой передано параметром

help %1 > help.tmp

rem Загружаем файл описания в редактор Блокнот

notepad help.tmp

Пусть командному файлу присвоили имя show-help.bat. Для того чтобы загрузить в блокнот описание команды, например, dir, нужно ввести команду следующим образом:

show-help.bat dir. Следующий командный файл создает каталог с именем, указанным в первом параметре, и записывает в него файл с текстом описания команды, указанной во втором параметре:

rem Пример командного файла с двумя параметрами

rem Создаем каталог с именем, заданным первым параметром

md %1

rem Создаем в нем файл с описанием команды,

rem заданной вторым параметром

help %2 > %1\%2.help

Что произойдет, если пользователь при запуске этого командного файла укажет не два, а четыре параметра? Ничего страшного, они ничему не помешают, просто не будут использованы. А что будет, если пользователь укажет только первый параметр? Второй параметр окажется пустым. Эффект получится такой: командный файл будет выполнен, но так, как будто на месте записи %2 ничего нет. Команда help сформирует список всех команд и поместит его в файл с пустым именем и расширением.help. Если же пользователь запустить этот файл, не указав ни одного параметра, то при попытке командного процессора выполнить команду md (напомним, она предназначена для создания каталога), мы получим сообщение о синтаксической ошибке, поскольку у команды md обязательно должен быть параметр. Таким образом, использование параметров создает большие возможности, но может существенно усложнить дело. Для того чтобы командный файл всегда работал корректно, необходимо проверять корректность указания пользователем параметров и каким-то образом реагировать на неполные или неверные входные данные. Можно, конечно, этого и не делать, но некорректно работающий командный файл может наломать дров, особенно, если он предусматривает удаление или перезапись данных.
Переменные

Переменной называется поименованное значение. В учебниках по программированию переменную обычно сравнивают с конвертом, на котором написано имя. Внутрь конверта можно положить нечто, например, определенную сумму денег — это ее значение. Как и в случае с конвертом, значение переменной можно изменить. Для объявления переменной и одновременно для присвоения ей значения применяется команда set. Пример записи этой команды показан далее:

rem Компилятор help-файлов в формате CHM

set help_compiler=c:\HTML Help Workshop\hcc.exe

Для извлечения значения переменной ее имя помещают между двумя знаками процента, как показано ниже:

rem Компилятор help-файлов в формате CHM
set help_compiler=c:\HTML Help Workshop\hcc.exe

rem Проект help-файла модуля "Склад"

set store_hpj=help\sources\store\store.hpj

rem Проект help-файла модуля "Продажи"

set sales_hpj=help\sources\sales\sales.hpj

rem Компилируем help-файлы

%help_compiler% %store_hpj%

%help_compiler% %sales_hpj%

Приведенная программа показывает, в чем польза переменных. Во-первых, они позволяют вместо длинного фрагмента (например, пути к компилятору help-файлов) использовать внутри командного файла синонимичный ему короткий. Как минимум, это удобно. Во-вторых, они позволяют избежать повторения в тексте командного файла фрагментов, которые в дальнейшем могут меняться. Представьте себе, что мы переустановили Microsoft HTML Workshop в другой каталог. Если в командном файле для записи пути к нему применяется переменная, то будет достаточно исправить в командном файле только одну строку, а именно, ту в которой присваивается значение переменной help_compiler. Если бы мы писали путь к компилятору всякий раз, когда его необходимо взывать, то после изменения пути нам пришлось бы исправить каждую такую строку. В приведенном примере их две, но в реальном проекте их с таким же успехом может оказаться пять или пятнадцать, по числу help-файлов, которые мы хотим скомпилировать. Проблема не в том, что исправлять вручную каждую строчку тяжело (в конце концов, команды “copy” и “paste” никто не отменял), а в том, что это сильно повышает вероятность случайной ошибки. Параметры командного файла тоже являются переменными, но от обычных переменных отличаются тем, что их значения задаются при запуске командного файла. При написании командных файлов часто применяют следующий прием: несколько значений переменных указывают рядом (или перемежая их какими-либо символами или строками), так, чтобы получить некоторое новое осмысленное значение. Пример приведен ниже: 

rem Путь к компилятору хелп-файлов

set help_compiler="c:\Program Files\HTML Help Workshop\hhc.exe"

rem Путь к каталогу, в котором находятся проекты хелп-файлов

set project_path=e:\work\projects\help-projects

rem Вызываем компилятор для обработки конкретного проекта,

rem имя которого передаем в первом параметре

%help_compiler% %project_path%\%1.hpj
Условные операторы
Команда if позволяет выделять в командном файле группы команд, которые выполняются или не выполняются в зависимости от определенных условий. Для чего это нужно? Проверка условия — почти необходимая мера при создании командных файлов, использующих параметры. Перед тем, как начинать работу, командный файл, вообще говоря, должен удостовериться в том, что ему передан корректный набор параметров. В противном случае велик риск, что он выполнится неверно или безрезультатно, а пользователю останется только гадать, в чем же проблема. Более того, если командный файл удаляет, перемещает или перезаписывает какие-либо данные, то при некорректных параметрах он может даже нанести ущерб. Далее показан командный файл компиляции help-файла. В начало командного файла добавлена проверка первого параметра на непустоту. Обратите внимание на такую особенность синтаксиса: для операции сравнения используется сдвоенный знак равенства. Если первый параметр оказывается непустым, срабатывает команда goto, которая «перебрасывает» командный процессор к указанной метке. В данном случае имя этой метки compile. Обратите внимание, что там, где метка находится, ее имя предваряется двоеточием, а в команде goto нет. При пустом первом параметре командный процессор переходит к следующей строке, которая выдает сообщение об ошибке. А потом к следующей, которая перебрасывает его в самый конец файла к метке с именем finish:

rem Проверяем, задан ли параметр

if not "%1"=="" goto compile

rem Если параметр пуст, выдаем сообщение об ошибке

echo Не указано имя проекта хелп-файла

rem и переходим в конец командного файла

rem к метке finish

goto finish

rem Это метка с именем compile

:compile

rem Ниже расположены команды компиляции

rem Путь к компилятору хелп-файлов

set help_compiler="c:\Program Files\HTML Help Workshop\hhc.exe"

rem Путь к каталогу, в котором находятся проекты хелп-файлов

set project_path=e:\work\projects\help-projects

rem Вызываем компилятор для обработки конкретного проекта,

rem имя которого передаем в первом параметре

%help_compiler% %project_path%\%1.hpj

rem Это метка с именем finish

:finish

Предложенный способ проверки параметра не самый удачный. Во-первых, если пользователь по ошибке укажет в качестве параметра имя несуществующего файла, командный файл этим удовлетворится и предпримет попытку компиляции. Более правильный способ — проверить, существует ли такой файл в действительности. Для этого в языке команд MS-DOS предусмотрено специальное слово exist. Поэтому лучше было бы написать: if exist %1.hpj goto compile. Во-вторых, активное использование команды goto (т.н. безусловного перехода) и меток сильно запутывают код. Технически они ничем не плохи, но отлаживать и сопровождать командный файл, написанный в таком стиле, довольно неудобно. Поэтому программисты издавна считают безусловный переход приемом нежелательным. Ниже показан более правильный, с точки зрения стиля программирования, структурированный вариант, в котором используется конструкция if…else. Работает она так: если условие истинно, выполняются команды в скобках после if, а если ложно, то в скобках после else.

@echo off

rem Проверяем, задан ли параметр

if not exist %1.hpj (

   rem Если параметр пуст, выдаем сообщение об ошибке

   echo Такого проекта хелп-файла не существует.
)else (

   rem Ниже расположены команды компиляции

   rem Путь к компилятору хелп-файлов

   set help_compiler="c:\Program Files\HTML Help Workshop\hhc.exe"

   rem Путь к каталогу, в котором находятся проекты хелп-файлов

   set project_path=e:\work\projects\help-projects

   rem Вызываем компилятор для обработки конкретного проекта,

   rem имя которого передаем в первом параметре

   %help_compiler% %project_path%\%1.hpj

)

Приведем еще один пример работы с проверками. Следующий командный файл создает каталог с именем help-files (предположим, для выгрузки в него скомпилированных help-файлов). При этом, если каталог с таким именем уже существует (и в нем, вероятно, находятся старые help-файлы, которые не хотелось бы терять: вдруг новые окажутся хуже?), командный файл присваивает ему расширение bak. Но если каталог help-files.bak уже существовал, то командный файл его удаляет (будем считать, что одной резервной копии нам хватит):

if exist help-files.bak rd help-files.bak

if exist help-files ren help-files help-files.bak

md help-files
Массовая обработка файлов

Команда for позволяет организовать выполнение повторяющихся однотипных действий. Можно использовать ее для того, чтобы вывести на экран числа от одного до десяти, как показано ниже: 

for /l %%i in (1,1,10) do echo %%i

Переменная i называется счетчиком цикла. В силу своеобразия синтаксиса команды for, имя счетчика цикла должно состоять из одной буквы. Причем, если мы пишем командный файл, то перед именем счетчика цикла надо поставить сдвоенный знак процента, если же мы просто набираем команду в командной строке, то одиночный. Логика работы этой команды такова. После слова in указан диапазон изменения счетчика цикла. В данном варианте команды это тройка чисел: начальное значение счетчика, шаг счета, предельное значение счетчика. При выполнении команды командный процессор сначала присвоит переменной i значение 1, а потом на каждом шаге цикла будет увеличивать его на 1, пока оно не превысит 10. Очевидно, таких шагов получится десять. Если бы в качестве шага счета мы указали число 2, то цикл выполнился бы пять раз. На каждом шаге цикла выполняется тело цикла, написанное после слова do. В приведенном примере это команда echo, которая выводит на экран текущее значение счетчика цикла. Наверно можно придумать ситуацию, когда что-то подобное на самом деле требуется, но обычно команда for используется для перебора и обработки файлов. Надо сказать, что в достаточно простых случаях массовая обработка файлов выполняется с помощью подстановочных символов. Если, мы хотим всем файлам в текущем каталоге заменить расширение .htm на .html, мы вводим команду ren *.htm *.html. Но если то же самое надо сделать не в одном каталоге, а в дереве каталогов, то без команды for не обойтись. Приведенный ниже командный файл выполняет эту операцию для всех htm-файлов в подкаталоге website текущего каталога. Точнее, во всем дереве каталогов, которое находится внутри website:

  for /r website %%i in (*.htm) do ren %%i %%~ni.html
Ключ /r указывает на необходимость обхода каталога website и всех его внутренностей. Если его не указать (но тогда и каталог указывать не разрешается), то обработаны будут только файлы в текущем каталоге. Диапазоном значений счетчика цикла в данном варианте команды является множество всех файлов с расширением .htm, находящихся внутри каталога (точнее, дерева) website. Странная на первый взгляд запись ~ni означает, что из значения переменной i требуется выделить только имя файла. В языке команд MS-DOS предусмотрено несколько таких модификаторов, например, запись ~хi обозначает расширение файла. Все модификаторы описаны в справке по команде for. Тело цикла может состоять из нескольких команд, заключенных в скобки:

@echo off

for /r website %%i in (*.htm) do (

   rem Выводим имя файла

   echo %%i

   rem Переименовываем файл

   ren %%i %%~ni.html

)

Передача управления другому командному файлу

Существует возможность вызвать из одного командного файла другой командный файл. Для этого служит команда call. Замечательно что, переменные, заданные в вызывающем командном файле «видны» вызванному. И наоборот, после того, как вызванный файл закончит работу и вернет управление вызвавшему, последний будет «видеть» переменные, оставленные ему вызванным «в наследство». Это позволяет разработчику командных файлов действовать, например, следующим образом. Если несколько командных файлов должны пользоваться одними и теми же значениями, допустим, путями к каким-то файлам, их можно вынести в отдельный командный файл, который будет играть роль конфигурационного файла. Каждый рабочий командный файл будет начинаться вызовом конфигурационного. Выигрыш в том, что при изменении путей вносить изменения придется только в один конфигурационный файл, а не во множество рабочих.

«Конфигурационный» командный файл config.bat:

rem Путь к компилятору хелп-файлов

set help_compiler="c:\Program Files\HTML Help Workshop\hhc.exe"

rem Путь к каталогу, в котором находятся проекты хелп-файлов

set project_path=e:\work\projects\help-projects

«Рабочий» командный файл:

@echo off

rem Настраиваем переменные

call config.bat

rem Проверяем, задан ли параметр

if not exist %1.hpj (

   rem Если параметр пуст, выдаем сообщение об ошибке

   echo Такого проекта хелп-файла не существует.

) else (

   rem Ниже расположены команды компиляции

   rem Вызываем компилятор для обработки конкретного проекта,

   rem имя которого передаем в первом параметре

   %help_compiler% %project_path%\%1.hpj

)

Вопросы для самопроверки

1. Что такое командные файлы?
2. Где появились впервые пакетные файлы?

3. Что такое командный процессор?
4. При описании параметров что пишетс в начале, а что в конце?

5. Чем отличаются абсолютный и относительный пути?

6. С помощью какого символа одну команду можно расположить на нескольких строчках

7. Что такое переменная?

8. Какая команда используется для объявления переменной?

Упражнения для самопроверки
1. Записать в файл copy.txt информацию о функции copy.
2. Существует каталог файлов. Требуется переместить все файлы из этого каталога с расширение .jpg в папку C:\documents\pictures\
3. Сущечтвует каталок файлов вывести имена файлов с расширением .doc
4. Существует каталог файлов. Требуется переименовать все файлы этого каталога с расширением .pas в .dpr . 

